Mat 1. 2-timersprøve den 17. maj 2016.

JE 13.5.16

Opgave 1

```
> restart;
```

Givet funktionen

```
> f:=x->sqrt(2*x-1);
```

$$f := x \to \sqrt{2\,x-1}$$

Spørgsmål 1

Funktionen er defineret for $2\,x-1 \geq 0 \Leftrightarrow x \geq \frac{1}{2}$. Dvs. Dm(f) er intervallet $[\frac{1}{2};\infty[$.

Spørgsmål 2

Med udviklingspunktet $x_0 = 1$ fås

```
> P3:=unapply(mtaylor(f(x),x=1,4),x);
```

$$P3 := x \to x - \frac{1}{2}\,(x-1)^2 + \frac{1}{2}\,(x-1)^3$$

Spørgsmål 3

```
> diff(f(x),x,x,x,x);
```

$$-\frac{15}{(2\,x-1)^{7/2}}$$

Ifølge Taylors formel med $x_0 = 1$ findes for ethvert $x \geq \frac{1}{2}$ et ξ mellem x og 1, så

$f(x) = P_3(x) + R_3(x)$, hvor

$f(x) - P_3(x) = R_3(x)$

$$= \frac{f^{(4)}(\xi)}{4!}\,(x-1)^4 = -\frac{15}{4!\,(2\,\xi-1)^{7/2}}\,(x-1)^4 = -\frac{5}{8}\cdot\frac{1}{(2\,\xi-1)^{7/2}}\,(x-1)^4 \;.$$

For $x = \frac{3}{2}$ findes da et ξ mellem $\frac{3}{2}$ og 1, så

$$f\left(\frac{3}{2}\right) = P_3\left(\frac{3}{2}\right) - \frac{5}{8}\cdot\frac{1}{(2\,\xi-1)^{7/2}}\left(\frac{3}{2}-1\right)^4 = P_3\left(\frac{3}{2}\right) - \frac{5}{8}\cdot\frac{1}{(2\,\xi-1)^{7/2}}\left(\frac{1}{2}\right)^4.$$

Benyttes $P_3\left(\frac{3}{2}\right)$ i stedet for $f\left(\frac{3}{2}\right)$, så er fejlen

$$\left|f\left(\frac{3}{2}\right) - P_3\left(\frac{3}{2}\right)\right| = \left|R_3\left(\frac{3}{2}\right)\right| = \frac{5}{8}\cdot\frac{1}{(2\,\xi-1)^{7/2}}\left(\frac{1}{2}\right)^4 \leq \frac{5}{8\cdot 2^4} = \frac{5}{2^7}\,,$$

da $1 \leq \xi \leq \frac{3}{2}$.

```
> P3(3/2);
```

$$\frac{23}{16}$$

```
> evalf(P3(3/2));
```

$$1.437500000$$

```
> evalf(5/2^7);
```

$$0.03906250000$$

```
> evalf(P3(3/2)-5/2^7);
```

$$1.398437500$$

Tilføjelse til spørgsmål 3:

Da $f\left(\frac{3}{2}\right) - P_3\left(\frac{3}{2}\right) = R_3\left(\frac{3}{2}\right) < 0$ fås

$P_3\left(\frac{3}{2}\right) - \frac{5}{2^7} \le f\left(\frac{3}{2}\right) < P_3\left(\frac{3}{2}\right)$ og dermed

$\frac{23}{16} - \frac{5}{2^7} \le f\left(\frac{3}{2}\right) < \frac{23}{16}$.

Altså : $1.3984 < f\left(\frac{3}{2}\right) = \sqrt{2} < 1.4375$.

Opgave 2

```
> restart;with(LinearAlgebra):with(plots):
```

Givet den symmetriske matrix

```
> A:=<<288/25,84/25>|<84/25,337/25>>;
```

$$A := \begin{bmatrix} \frac{288}{25} & \frac{84}{25} \\ \frac{84}{25} & \frac{337}{25} \end{bmatrix}$$

Spørgsmål 1

```
> g:=(x,y)->evalb(x[1]<y[1]):
> ev:=sort(Eigenvectors(A,output=list),g);
```

$$ev := \left[\left[9, 1, \left\{\begin{bmatrix} -\frac{4}{3} \\ 1 \end{bmatrix}\right\}\right], \left[16, 1, \left\{\begin{bmatrix} \frac{3}{4} \\ 1 \end{bmatrix}\right\}\right]\right]$$

Heraf aflæses, at $\mathbf{A} \in \mathbb{R}^{2\times 2}$ har egenværdierne 9 og 16 med $E_9 = \text{span}\left\{\begin{bmatrix} -\frac{4}{3} \\ 1 \end{bmatrix}\right\}$ og

$E_{16} = \text{span}\left\{\begin{bmatrix} \frac{3}{4} \\ 1 \end{bmatrix}\right\}$.

Da $\mathbf{A}$ er symmetrisk er de to egenvektorrum E_9 og E_{16} ortogonale.

```
> v1:=ev[1,3,1];
```

$$v1 := \begin{bmatrix} -\frac{4}{3} \\ 1 \end{bmatrix}$$

er en basis for E_9 .

```
> q1:=(-1)*Normalize(v1,Euclidean);
```

$$q1 := \begin{bmatrix} \frac{4}{5} \\ -\frac{3}{5} \end{bmatrix}$$

er en ortonormal basis for E_9 .

```
> v2:=ev[2,3,1];
```

$$v2 := \begin{bmatrix} \frac{3}{4} \\ 1 \end{bmatrix}$$

er en basis for E_{16} .

```
> q2:=Normalize(v2,Euclidean);
```

$$q2 := \begin{bmatrix} \frac{3}{5} \\ \frac{4}{5} \end{bmatrix}$$

hvor $\mathbf{q}_2 = \hat{\mathbf{q}}_1$,er en ortonormal basis for E_{16} .

Egenvektorerne $(\mathbf{q}_1, \mathbf{q}_2)$ er da en ortonormal basis for $\mathbb{R}^2$ udstyret med det sædvanlige skalarprodukt.

Sættes

```
> Q:=<q1|q2>;
```

$$Q := \begin{bmatrix} \frac{4}{5} & \frac{3}{5} \\ -\frac{3}{5} & \frac{4}{5} \end{bmatrix}$$

og

```
> Lambda:=DiagonalMatrix([9,16]);
```

$$\Lambda := \begin{bmatrix} 9 & 0 \\ 0 & 16 \end{bmatrix}$$

så er $\mathbf{Q}$ positiv ortogonal og

$$\mathbf{\Lambda} = \mathbf{Q}^{-1} \mathbf{A}\, \mathbf{Q} = \mathbf{Q}^{\mathrm{T}} \mathbf{A}\, \mathbf{Q}$$

Kontrol med Maple:

```
> Transpose(Q).Q;Determinant(Q);
```

$$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

$$1$$

```
> Q^(-1).A.Q;
```

$$\begin{bmatrix} 9 & 0 \\ 0 & 16 \end{bmatrix}$$

```
> Transpose(Q).A.Q;
```

$$\begin{bmatrix} 9 & 0 \\ 0 & 16 \end{bmatrix}$$

Spørgsmål 2

En ellipse E er i et sædvanligt retvinklet (x, y)-koordinatsystem i planen givet ved matrixligningen

$$\begin{bmatrix} x & y \end{bmatrix} \mathbf{A} \begin{bmatrix} x \\ y \end{bmatrix} = 144 .$$

Af spørgsmål 1 haves nu følgende:
$(\mathrm{O}\,;\, \mathbf{q}_1, \mathbf{q}_2)$ er et nyt sædvanligt retvinklet koordinatsystem i planen fremkommet ved en drejning af det givne koordinatsystem om O.
Forbindelsen mellem de gamle koordinater (x, y) og de nye koordinater (x_1, y_1) for et punkt P i planen er givet ved

$$\begin{bmatrix} x \\ y \end{bmatrix} = \mathbf{Q} \begin{bmatrix} x_1 \\ y_1 \end{bmatrix} .$$

Ellipsens ligning i det nye koordinatsystem er

$$\begin{bmatrix} x_1 & y_1 \end{bmatrix} \mathbf{Q}^{\mathrm{T}} \mathbf{A}\, \mathbf{Q} \begin{bmatrix} x_1 \\ y_1 \end{bmatrix} = 144 \Leftrightarrow \begin{bmatrix} x_1 & y_1 \end{bmatrix} \begin{bmatrix} 9 & 0 \\ 0 & 16 \end{bmatrix} \begin{bmatrix} x_1 \\ y_1 \end{bmatrix} = 144 \Leftrightarrow 9\, x_1^{\,2} + 16\, y_1^{\,2} = 144 \Leftrightarrow$$

$$\frac{x_1^{\,2}}{16} + \frac{y_1^{\,2}}{9} = 1.$$

Heraf aflæses, at ellipsens halve storakse a = 4 og ellipsens halve lilleakse b = 3.

X_1-aksen, som er linien gennem O med retningsvektor $\mathbf{q}_1$, har ligningen $y = -\frac{3}{4} x$.

Y_1-aksen, som er linien gennem O med retningsvektor $\mathbf{q}_2$, har ligningen $y = \frac{4}{3} x$.

```
> <x|y>.A.<x,y>;
```

$$\left(\frac{288}{25} x + \frac{84}{25} y\right) x + \left(\frac{84}{25} x + \frac{337}{25} y\right) y$$

```
> E:=expand(%)=144;
```

$$E := \frac{288}{25}\,x^2 + \frac{168}{25}\,x\,y + \frac{337}{25}\,y^2 = 144$$

```
> implicitplot({E,y=-3/4*x,y=4/3*x},x=-5..5,y=-5..5,scaling=
  constrained);
```



Opgave 3

```
> restart;with(LinearAlgebra):with(plots):
```

For en glat funktion $f : \mathbb{R}^2 \rightarrow \mathbb{R}$ med $f(0,0) = 0$ er et vektorfelt $\mathbf{V}$ i (x, y) - planen givet ved $\mathbf{V}(x, y) = \nabla f(x, y) = (f'_x(x,y)\ ,\ f'_y(x,y)) = (x - y^2 + 1, -2\,xy)$

Spørgsmål 1

$\nabla f(x, y) = (0, 0) \Leftrightarrow x = 0$ og $y = \pm 1$ eller $y = 0$ og $x = -1$.
Samtlige stationære punkter for f er da $(0, 1)$, $(0, -1)$ *og* $(-1, 0)$.

Spørgsmål 2

Hvis f har egentligt lokalt maksimum eller egentligt lokalt minimum i et punkt, så må punktet være et stationært punkt, da f ikke har undtagelsespunkter.

```
> diff(x-y^2+1,x);
```

$$1$$

```
> diff(-2*x*y,y);
```

$$-2\,x$$

```
> diff(x-y^2+1,y);
```

$$-2\,y$$

Hessematricen for f i punktet (x, y) er

```
> H(x,y):=<<1,-2*y>|<-2*y,-2*x>>;
```

$$H(x, y) := \begin{bmatrix} 1 & -2\,y \\ -2\,y & -2\,x \end{bmatrix}$$

```
> H(0,1):=subs(x=0,y=1,H(x,y));
```

$$H(0, 1) := \begin{bmatrix} 1 & -2 \\ -2 & 0 \end{bmatrix}$$

```
> Eigenvalues(H(0,1),output=list);
```

$$\left[\frac{1}{2} + \frac{1}{2}\sqrt{17}, \frac{1}{2} - \frac{1}{2}\sqrt{17}\right]$$

Da de to egenværdier for $\mathbf{H}(0, 1)$ har modsat fortegn, så har f hverken egentligt lokalt maksimum eller egentligt lokalt minimum i det stationære punkt $(0, 1)$.

```
> H(0,-1):=subs(x=0,y=-1,H(x,y));
```

$$H(0, -1) := \begin{bmatrix} 1 & 2 \\ 2 & 0 \end{bmatrix}$$

```
> Eigenvalues(H(0,-1),output=list);
```

$$\left[\frac{1}{2} + \frac{1}{2}\sqrt{17}, \frac{1}{2} - \frac{1}{2}\sqrt{17}\right]$$

Da de to egenværdier for $\mathbf{H}(0,-1)$ har modsat fortegn, så har f hverken egentligt lokalt maksimum eller egentlig lokalt minimum i det stationære punkt $(0,-1)$.

```
> H(-1,0):=subs(x=-1,y=0,H(x,y));
```

$$H(-1, 0) := \begin{bmatrix} 1 & 0 \\ 0 & 2 \end{bmatrix}$$

```
> Eigenvalues(H(-1,0),output=list);
```

$$[2, 1]$$

Da begge egenværdier for $\mathbf{H}(-1, 0)$ er positive, har f egentligt lokalt minimum i det stationære punkt $(-1, 0)$.

Altså:

f har netop ét egentligt lokalt minimum nemlig i punktet $(-1, 0)$ og ingen egentlige lokale maksima.

Spørgsmål 3

Trappelinie K: $(0, 0) \rightarrow (x, 0) \rightarrow (x, y)$.

Da gradientfeltet $\nabla f = (V_x, V_y)$ er givet, kan vi finde f med $f(0, 0) = 0$ ved formlen

$$f(x, y) = f(0, 0) + \mathrm{Tan}(\nabla f, K) = \mathrm{Tan}(\nabla f, K) = \int_0^x V_x(t, 0)\,\mathrm{d}t + \int_0^y V_y(x, t)\,\mathrm{d}t$$

$$= \int_0^x (t+1)\,\mathrm{d}t + \int_0^y -2\,xt\,\mathrm{d}t = \frac{1}{2}x^2 + x - xy^2.$$

Spørgsmål 4

```
> f:=(x,y)->1/2*x^2+x-x*y^2;
```

$$f := (x, y) \rightarrow \frac{1}{2}\,x^2 + x - x\,y^2$$

```
> 'f(-1,0)'=f(-1,0);
```

$$f(-1, 0) = -\frac{1}{2}$$

```
> contourplot(f(x,y),x=-3..3,y=-2..2,contours=25);
```



Opgave 4

```
> restart;with(LinearAlgebra):with(plots):
> prik:=(x,y)->VectorCalculus[DotProduct](x,y):
> kryds:=(x,y)->convert(VectorCalculus[CrossProduct](x,y),Vector):
```

$A = \{(x, y) | \ 0 \leq x \leq 2 \text{ og } -\frac{\pi}{2} \leq y \leq \frac{\pi}{2}\}.$

Vi betragter funktionen

```
> h:=(x,y)->x*cos(y);
```

$$h := (x, y) \rightarrow x \cos(y)$$

for $(x, y) \in A$.

Graffladen $F = \{(x, y, z) \mid 0 \leq x \leq 2, -\frac{\pi}{2} \leq y \leq \frac{\pi}{2}, z = h(x, y)\}$.

```
> plot3d(h(x,y),x=0..2,y=-Pi/2..Pi/2,scaling=constrained,view=0.
  .2);
```



Spørgsmål 1

Parameterfremstilling for F

```
> r:=<u,v,h(u,v)>;
```

$$r := \begin{bmatrix} u \\ v \\ u\cos(v) \end{bmatrix}$$

hvor $u \in [0; 2]$ og $v \in [-\frac{\pi}{2}; \frac{\pi}{2}]$.

```
> r1:=diff~(r,u);
```

$$r1 := \begin{bmatrix} 1 \\ 0 \\ \cos(v) \end{bmatrix}$$

```
> r2:=diff~(r,v);
```

$$r2 := \begin{bmatrix} 0 \\ 1 \\ -u\sin(v) \end{bmatrix}$$

Fladens normalvektor er

```
> N:=kryds(r1,r2);
```

$$N := \begin{bmatrix} -\cos(v) \\ u\sin(v) \\ 1 \end{bmatrix}$$

Spørgsmål 2

I dette spørgsmål og i det næste spørgsmål betragtes desuden et vektorfelt **V** i (x, y, z) - rummet om hvilket det oplyses, at

$\mathrm{Div}(\mathbf{V})(x, y, z) = x + y + z$ og $\mathbf{Rot}(\mathbf{V})(x, y, z) = (3z, 3x, 3y)$.

Med den valgte orientering af den lukkede randkurve ∂F for F er

$\mathbf{n}_F = \frac{\mathbf{N}(u, v)}{|\mathbf{N}(u, v)|}$ (højrekonvention).

Af Stokes' sætning fås da

$$\mathrm{Cirk}(\mathbf{V}, \partial F) = \mathrm{Flux}(\mathbf{Rot}(\mathbf{V}), F) = \int_F \mathbf{n}_F \cdot \mathbf{Rot}(\mathbf{V})\, \mathrm{d}\mu = \int_{-\frac{\pi}{2}}^{\frac{\pi}{2}} \int_0^2 \mathbf{N}(u, v) \cdot \mathbf{Rot}(\mathbf{V})(\mathbf{r}(\mathrm{u,v}))\mathrm{d}u\, \mathrm{d}v$$

```
> integrand:=prik(N,<3*u*cos(v),3*u,3*v>);
```

$$integrand := -3\cos(v)^2 u + 3u^2 \sin(v) + 3v$$

```
> Int(Int(integrand,u=0..2),v=-Pi/2..Pi/2)=int(int(integrand,u=
  0..2),v=-Pi/2..Pi/2);
```

$$\int_{-\frac{1}{2}\pi}^{\frac{1}{2}\pi} \int_0^2 \left(-3\cos(v)^2 u + 3u^2\sin(v) + 3v\right) \mathrm{d}u\, \mathrm{d}v = -3\pi$$

Spørgsmål 3

Parameterfremstilling for det massive område Ω i rummet

```
> R:=<u,v,w*u*cos(v)>;
```

$$R := \begin{bmatrix} u \\ v \\ w\, u \cos(v) \end{bmatrix}$$

hvor $u \in [0; 2]$, $v \in [-\frac{\pi}{2}; \frac{\pi}{2}]$ og $w \in [0; 1]$.

∂Ω er den lukkede overflade af Ω orienteret med udadrettet enhedsnormalvektor.
Af Gauss' sætning fås da

$$\text{Flux}(\mathbf{V}, \partial\Omega) = \int_{\Omega} \text{Div}\,(\mathbf{V})\, d\mu = \int_0^1 \int_{-\frac{\pi}{2}}^{\frac{\pi}{2}} \int_0^2 \text{Div}\,(\mathbf{V})(\mathbf{R}(u,v,w))\, |\text{JR}(u,v,w)|\, du\, dv\, dw$$

```
> M:=<diff~(R,u)|diff~(R,v)|diff~(R,w)>;
```

$$M := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ w \cos(v) & -w\, u \sin(v) & u \cos(v) \end{bmatrix}$$

```
> JR:=Determinant(M);
```

$$JR := u \cos(v)$$

som er ≥ 0 , da $u \in [0; 2]$ og $v \in [-\frac{\pi}{2}; \frac{\pi}{2}]$.

```
> integrand:=expand((w*u*cos(v)+v+u)*JR);
```

$$integrand := u^2 \cos(v)^2 w + u \cos(v)\, v + u^2 \cos(v)$$

```
> Int(Int(Int(integrand,u=0..2),v=-Pi/2..Pi/2),w=0..1)=int(int
  (int(integrand,u=0..2),v=-Pi/2..Pi/2),w=0..1);
```

$$\int_0^1 \int_{-\frac{1}{2}\pi}^{\frac{1}{2}\pi} \int_0^2 \left(u^2 \cos(v)^2 w + u \cos(v)\, v + u^2 \cos(v) \right) du\, dv\, dw = \frac{2}{3}\pi + \frac{16}{3}$$